

## Textový súbor

Súbor je pomenovaná skupina údajov. Textový súbor pozostáva iba zo znakov. Ak potrebujeme v programe spracúvať väčšie množstvo údajov, je výhodnejšie, presnejšie a rýchlejšie načítať ich z textového súboru. Niekedy je zasa výhodné vypisovať údaje nielen na obrazovku, ale aj do súboru.

Keďže súbor je uložený na vonkajšom pamäťovom médiu a nie v operačnej pamäti, ako bežiaci program, musíme jeho obsah sprístupniť programu. To sa deje cez vyrovnávacie pamäte. Vyrovnávacia pamäť – buffer, je premenná typu **text** pre sprístupnenie obsahu textového súboru. Túto vyrovnávaciu pamäť deklarujeme príkazom:

```
premenné : text ;
```

Program môže pracovať naraz s viacerými súbormi. Preto je nutné priradiť premennej typu text príslušný súbor. Jednej premennej môže byť priradený len jeden súbor! Na to slúži príkaz:

```
assign(premenená , názov_súboru) ;
```

Súbor môžeme otvoriť:

- pre čítanie - príkazom: **reset**(**premenná**) ;
- pre zápis – **rewrite**(**premenná**) ;
- pre dopisovanie na koniec súboru – **append**(**premenná**) ;

Po otvorení súboru z disku pre čítanie, sa do vyrovnávacej pamäte načíta obsah súboru. Ten sa potom operáciami vstupu `read` a `readln` načítava do jednotlivých premenných. Ak chceme procedúrami `read` a `readln` čítať údaje zo súboru, tak prvým parametrom pri ich volaní musí byť premenná vyrovnávacej pamäte príslušného súboru.

Po otvorení súboru pre zápis je vyrovnávacia pamäť prázdna. Operáciami `write` a `writeln`, sa zapisujú údaje do vyrovnávacej pamäte. Ak chceme procedúrami `write` a `writeln` zapisovať údaje do súboru, tak prvým parametrom pri ich volaní musí byť premenná vyrovnávacej pamäte príslušného súboru. Obsahom vyrovnávacej pamäte sa prepíše obsah súboru, prípadne sa vytvorí nový súbor, ak takýto súbor neexistuje, až keď v programe **súbor zatvoríme!** Súbor zatvoríme príkazom **close**(**premenná**) ; Aj súbor otvorený pre čítanie je nutné pred ukončením činnosti programu zatvoriť, inak by nám tento nedostatok mohol neskôr spôsobiť zamrznutie operačného systému. Pamätajte si, že **každý otvorený súbor treba zatvoriť!**

Veľmi užitočnou funkciou pri práci so súbormi je funkcia **eof**(**premenná**) ; Táto funkcia vráti hodnotu **true**, ak sme už prečítali zo súboru všetky údaje (súbor je prázdny). V opačnom prípade vráti hodnotu **false**.

Názov funkcie eof je skratka slov **end of file**, čo v preklade do slovenčiny znamená koniec súboru.

### Príklad 1:

Zostavte program, ktorý načíta zo súboru *vstup.txt* najprv počet čísel N a potom N reálnych čísel a vypíše na obrazovku načítané čísla, ich súčet a aritmetický priemer:

### Riešenie:

*Skôr, než začnete program ladiť, musíte si na disku v priečinku, do ktorého ste si ladený program uložili, vytvoriť textový súbor s názvom vstup.txt a napísať doň do jednotlivých riadkov (v jednom riadku jedno číslo) napríklad tieto čísla 5, 3.2, 6.9, 7.1, 8.5, 4.0.*

```
uses Crt;
var i,n:integer;
    a:array[1..20] of real;
    ap,sum:real;
    f:text;
begin
  clrscr;
  {citanie vstupnych udajov zo suboru do pola}
  assign(f,'vstup.txt');
  reset(f);
  readln(f,n);
  for i:=1 to n do readln(f,a[i]);
  close(f);

  {vypocty vysledkov}
  sum:=0;
  for i:=1 to n do sum:=sum+a[i];
  ap:=sum/n;

  {vypis obsahu pola a vysledkov na obrazovku}
  clrscr;
  writeln('Nacital som tieto cisla:');
  for i:=1 to n do write(a[i]:7:2,' ');
  writeln;
  writeln;
  writeln('Sucet zadanych cisel je ',sum:7:2);
  writeln('Aritmeticky priemer zadanych cisel je:
',ap:7:2);
  repeat until keypressed;
end.
```

### Príklad 2:

Zostavte program, ktorý načíta zo súboru *vstup.txt* reálne čísla, pričom nevie dopredu ich počet a vypíše do súboru *vystup.txt* načítané čísla, ich súčet a aritmetický priemer:

**Riešenie:**

```
uses Crt;
var i,n:integer;
    a:array[1..20] of real;
    ap,sum:real;
    f:text;
begin
  clrscr;
  {citanie vstupnych udajov zo suboru do pola}
  assign(f,'vstup.txt');
  reset(f);
  n:=0;
  while not eof(f) do
    begin
      inc(n);
      readln(f,a[n]);
    end;
  close(f);

  {vypocty vysledkov}
  sum:=0;
  for i:=1 to n do sum:=sum+a[i];
  ap:=sum/n;

  {vypis obsahu pola a vysledkov do suboru}
  clrscr;
  assign(f,'vystup.txt');
  rewrite(f);
  writeln(f,'Nacital som tieto cisla:');
  for i:=1 to n do write(f,a[i]:7:2,' ');
  writeln(f);
  writeln(f);
  writeln(f,'Sucet zadanych cisel je ',sum:7:2);
  writeln(f,'Aritmeticky priemer zadanych cisel je:
',ap:7:2);
  close(f);
  writeln('Hotovo!');
  repeat until keypressed;
end.
```